Lesson 15. Work Scheduling Models, Revisited

1 The postal workers problem, revisited

Example 1. Postal employees in Simplexville work for 5 consecutive days, followed by 2 days off, repeated weekly. Below are the minimum number of employees needed for each day of the week:

Day	Employees needed
Monday (1)	7
Tuesday (2)	8
Wednesday (3)	7
Thursday (4)	6
Friday (5)	6
Saturday (6)	4
Sunday (7)	5

We want to determine the minimum total number of employees needed.

Our original model:

Decision variables. Let

$$\begin{aligned} x_1 &= \text{number of employees who start work on Monday and work though Friday} \\ x_2 &= \text{number of employees who start work on Tuesday and work though Saturday} \\ &\vdots \\ x_7 &= \text{number of employees who start work on Sunday and work through Thursday} \end{aligned}$$

Objective function and constraints.

$$\begin{aligned} \min \quad & x_1 + x_2 + x_3 + x_4 + x_5 + x_6 + x_7 & & \\ \text{s.t.} \quad & x_1 + x_4 + x_5 + x_6 + x_7 \geq 7 & & \text{(Mon)} \\ & x_1 + x_2 + x_5 + x_6 + x_7 \geq 8 & & \text{(Tue)} \\ & x_1 + x_2 + x_3 + x_6 + x_7 \geq 7 & & \text{(Wed)} \\ & x_1 + x_2 + x_3 + x_4 + x_7 \geq 6 & & \text{(Thu)} \\ & x_1 + x_2 + x_3 + x_4 + x_5 \geq 6 & & \text{(Fri)} \\ & x_2 + x_3 + x_4 + x_5 + x_6 \geq 4 & & \text{(Sat)} \\ & x_3 + x_4 + x_5 + x_6 + x_7 \geq 5 & & \text{(Sun)} \\ & x_1, \; x_2, \; x_3, \; x_4, \; x_5, \; x_6, \; x_7 \geq 0 & & \end{aligned}$$

- Left hand side of (Mon): add up the variables x_i such that shift i covers Monday
- We need a way to specify elements of a set that meet certain characteristics

2 Some more set notation

- What if we only want certain elements of a set?
- " : " notation

$$j \in S : [\text{condition}] \quad \Leftrightarrow \quad j \in \text{elements of } S \text{ such that } [\text{condition}] \text{ holds}$$

- For example:
  - Define $N = \{1, 2, 3\}$, $S_1 = \{a, b\}$, $S_2 = \{b, c\}$, $S_3 = \{a, c\}$
  - Then
    - $j \in N : j \geq 2 \quad \Leftrightarrow$
    - $j \in N : a \in S_j \quad \Leftrightarrow$
- Some people use " | " instead ":"

Describe the input parameters of Example 1 using sets and for statements.

Write a linear program for Example 1 using the symbolic input parameters you described above.

3 GMPL code

GMPL model file for this linear program:

```
# Model for postal employees problem in Lesson 13

## Input parameters ##
set days;                       # days of the week
set shifts;                     # shifts
set shift_days{j in shifts};    # days worked for each shift
param required{i in days};      # number of employees needed for each day

## Decision variables and variable bounds ##
var x{j in shifts} >= 0;        # number of employees assigned to each shift

## Objective function ##
# Minimize total number of employees
minimize total_employees: sum{j in shifts} x[j];

## General constraints ##
# Number of employees working on day i >= minimum required on day i
subject to employees_needed{i in days}:
  sum{j in shifts: i in shift_days[j]} x[j] >= required[i];

end;
```

GMPL data file for this linear program:

```
# Days of the week
set days := Mon Tue Wed Thu Fri Sat Sun;

# Shifts
set shifts := 1 2 3 4 5 6 7;

# Days worked in each shift
set shift_days[1] := Mon Tue Wed Thu Fri;
set shift_days[2] := Tue Wed Thu Fri Sat;
set shift_days[3] := Wed Thu Fri Sat Sun;
set shift_days[4] := Thu Fri Sat Sun Mon;
set shift_days[5] := Fri Sat Sun Mon Tue;
set shift_days[6] := Sat Sun Mon Tue Wed;
set shift_days[7] := Sun Mon Tue Wed Thu;

# Number of employees needed for each day
param required :=
  Mon   7
  Tue   8
  Wed   7
  Thu   6
  Fri   6
  Sat   4
  Sun   5;

end;
```